pragma solidity ^0.4.18;


/**
 * @title ERC20Basic
 * @dev Simpler version of ERC20 interface
 * @dev see https://github.com/ethereum/EIPs/issues/179
 */
contract ERC20Basic {
  uint256 public totalSupply;
  function balanceOf(address who) public view returns (uint256);
  function transfer(address to, uint256 value) public returns (bool);
  event Transfer(address indexed from, address indexed to, uint256 value);
}



/**
 * @title Ownable
 * @dev The Ownable contract has an owner address, and provides basic authorization control
 * functions, this simplifies the implementation of "user permissions".
 */
contract Ownable {
  address public owner;


  event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);


  /**
   * @dev The Ownable constructor sets the original `owner` of the contract to the sender
   * account.
   */
  function Ownable() public {
    owner = msg.sender;
  }


  /**
   * @dev Throws if called by any account other than the owner.
   */
  modifier onlyOwner() {
    require(msg.sender == owner);
    _;
  }


  /**
   * @dev Allows the current owner to transfer control of the contract to a newOwner.
   * @param newOwner The address to transfer ownership to.
   */
  function transferOwnership(address newOwner) public onlyOwner {
    require(newOwner != address(0));
    OwnershipTransferred(owner, newOwner);
    owner = newOwner;
  }

}












/**
 * @title SafeMath
 * @dev Math operations with safety checks that throw on error
 */
library SafeMath {
  function mul(uint256 a, uint256 b) internal pure returns (uint256) {
    if (a == 0) {
      return 0;
    }
    uint256 c = a * b;
    assert(c / a == b);
    return c;
  }

  function div(uint256 a, uint256 b) internal pure returns (uint256) {
    // assert(b > 0); // Solidity automatically throws when dividing by 0
    uint256 c = a / b;
    // assert(a == b * c + a % b); // There is no case in which this doesn't hold
    return c;
  }

  function sub(uint256 a, uint256 b) internal pure returns (uint256) {
    assert(b <= a);
    return a - b;
  }

  function add(uint256 a, uint256 b) internal pure returns (uint256) {
    uint256 c = a + b;
    assert(c >= a);
    return c;
  }
}



/**
 * @title Basic token
 * @dev Basic version of StandardToken, with no allowances.
 */
contract BasicToken is ERC20Basic {
  using SafeMath for uint256;

  mapping(address => uint256) balances;

  /**
  * @dev transfer token for a specified address
  * @param _to The address to transfer to.
  * @param _value The amount to be transferred.
  */
  function transfer(address _to, uint256 _value) public returns (bool) {
    require(_to != address(0));
    require(_value <= balances[msg.sender]);

    // SafeMath.sub will throw if there is not enough balance.
    balances[msg.sender] = balances[msg.sender].sub(_value);
    balances[_to] = balances[_to].add(_value);
    Transfer(msg.sender, _to, _value);
    return true;
  }

  /**
  * @dev Gets the balance of the specified address.
  * @param _owner The address to query the the balance of.
  * @return An uint256 representing the amount owned by the passed address.
  */
  function balanceOf(address _owner) public view returns (uint256 balance) {
    return balances[_owner];
  }

}







/**
 * @title ERC20 interface
 * @dev see https://github.com/ethereum/EIPs/issues/20
 */
contract ERC20 is ERC20Basic {
  function allowance(address owner, address spender) public view returns (uint256);
  function transferFrom(address from, address to, uint256 value) public returns (bool);
  function approve(address spender, uint256 value) public returns (bool);
  event Approval(address indexed owner, address indexed spender, uint256 value);
}



/**
 * @title Standard ERC20 token
 *
 * @dev Implementation of the basic standard token.
 * @dev https://github.com/ethereum/EIPs/issues/20
 * @dev Based on code by FirstBlood: https://github.com/Firstbloodio/token/blob/master/smart_contract/FirstBloodToken.sol
 */
contract StandardToken is ERC20, BasicToken {

  mapping (address => mapping (address => uint256)) internal allowed;


  /**
   * @dev Transfer tokens from one address to another
   * @param _from address The address which you want to send tokens from
   * @param _to address The address which you want to transfer to
   * @param _value uint256 the amount of tokens to be transferred
   */
  function transferFrom(address _from, address _to, uint256 _value) public returns (bool) {
    require(_to != address(0));
    require(_value <= balances[_from]);
    require(_value <= allowed[_from][msg.sender]);

    balances[_from] = balances[_from].sub(_value);
    balances[_to] = balances[_to].add(_value);
    allowed[_from][msg.sender] = allowed[_from][msg.sender].sub(_value);
    Transfer(_from, _to, _value);
    return true;
  }

  /**
   * @dev Approve the passed address to spend the specified amount of tokens on behalf of msg.sender.
   *
   * Beware that changing an allowance with this method brings the risk that someone may use both the old
   * and the new allowance by unfortunate transaction ordering. One possible solution to mitigate this
   * race condition is to first reduce the spender's allowance to 0 and set the desired value afterwards:
   * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
   * @param _spender The address which will spend the funds.
   * @param _value The amount of tokens to be spent.
   */
  function approve(address _spender, uint256 _value) public returns (bool) {
    allowed[msg.sender][_spender] = _value;
    Approval(msg.sender, _spender, _value);
    return true;
  }

  /**
   * @dev Function to check the amount of tokens that an owner allowed to a spender.
   * @param _owner address The address which owns the funds.
   * @param _spender address The address which will spend the funds.
   * @return A uint256 specifying the amount of tokens still available for the spender.
   */
  function allowance(address _owner, address _spender) public view returns (uint256) {
    return allowed[_owner][_spender];
  }

  /**
   * approve should be called when allowed[_spender] == 0. To increment
   * allowed value is better to use this function to avoid 2 calls (and wait until
   * the first transaction is mined)
   * From MonolithDAO Token.sol
   */
  function increaseApproval(address _spender, uint _addedValue) public returns (bool) {
    allowed[msg.sender][_spender] = allowed[msg.sender][_spender].add(_addedValue);
    Approval(msg.sender, _spender, allowed[msg.sender][_spender]);
    return true;
  }

  function decreaseApproval(address _spender, uint _subtractedValue) public returns (bool) {
    uint oldValue = allowed[msg.sender][_spender];
    if (_subtractedValue > oldValue) {
      allowed[msg.sender][_spender] = 0;
    } else {
      allowed[msg.sender][_spender] = oldValue.sub(_subtractedValue);
    }
    Approval(msg.sender, _spender, allowed[msg.sender][_spender]);
    return true;
  }

}







/**
 * @title Destructible
 * @dev Base contract that can be destroyed by owner. All funds in contract will be sent to the owner.
 */
contract Destructible is Ownable {

  function Destructible() public payable { }

  /**
   * @dev Transfers the current balance to the owner and terminates the contract.
   */
  function destroy() onlyOwner public {
    selfdestruct(owner);
  }

  function destroyAndSend(address _recipient) onlyOwner public {
    selfdestruct(_recipient);
  }
}


contract GeocashToken is StandardToken, Destructible {
  string public name;
  string public symbol;
  uint public decimals;
  uint public buyPriceInWei;
  uint public sellPriceInWei;
  uint public minBalanceForAccounts;
  address public companyWallet;

  mapping(address => uint256) balances;
  mapping (address => bool) public frozenAccounts;
  event FrozenFunds(address target, bool frozen);

  event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

  modifier onlyOwner() {
    require(msg.sender == owner);
    _;
  }

  function GeocashToken(uint256 _minBalanceForAccountsInWei, uint256 _buyPriceInWei, uint256 _sellPriceInWei, address _companyWallet) public {
    name = 'Geocash';
    symbol = 'GEO';
    decimals = 18;
    totalSupply = 500000000 * (10 ** uint256(decimals));
    balances[this] = totalSupply;
    minBalanceForAccounts = _minBalanceForAccountsInWei;
    buyPriceInWei = _buyPriceInWei;
    sellPriceInWei = _sellPriceInWei;
    companyWallet = _companyWallet;
  }


  /*Constant functions*/
  function balanceOf(address _owner) public view returns (uint256 balance) {
    return balances[_owner];
  }

  function isFrozen(address _owner) public view returns (bool frozen){
    return frozenAccounts[_owner];
  }




  /*Non constant functions*/

  function transfer(address _to, uint256 _value) public returns (bool) {
    require(_to != address(0));
    require(!frozenAccounts[msg.sender]);
    require(!frozenAccounts[_to]);
    require(_value <= balances[msg.sender]);
    if(msg.sender.balance < minBalanceForAccounts){
      sell((minBalanceForAccounts.sub(msg.sender.balance)).div(sellPriceInWei));
    }
    balances[msg.sender] = balances[msg.sender].sub(_value);
    balances[_to] = balances[_to].add(_value);
    Transfer(msg.sender, _to, _value);
    return true;
  }

  function setBuyPrice(uint _buyPriceInWei) onlyOwner public returns (bool){
    require(_buyPriceInWei > 0);
    buyPriceInWei = _buyPriceInWei;
    return true;
  }

  function setSellPrice(uint _sellPriceInWei) onlyOwner public returns (bool){
    require(_sellPriceInWei > 0);
    sellPriceInWei = _sellPriceInWei;
    return true;
  }

  function setCompanyWallet(address _wallet) onlyOwner public returns (bool){
    require(_wallet != address(0));
    companyWallet = _wallet;
    return true;
  }

  function buy() public payable returns (uint){
    require(msg.sender != address(0));
    require(msg.value >= 0);
    uint amount = msg.value.div(buyPriceInWei).mul(1 ether);
    require(amount > 0);
    require(balances[this] >= amount);
    uint oldBalance = balances[this].add(balances[msg.sender]);
    balances[this] = balances[this].sub(amount);
    balances[msg.sender] = balances[msg.sender].add(amount);
    uint newBalance = balances[this].add(balances[msg.sender]);
    assert(newBalance == oldBalance);
    Transfer(this, msg.sender, amount);
    return amount;
  }

  function sell(uint _amount) internal returns(uint revenue) {
    require(_amount > 0);
    require(balances[msg.sender]>= _amount);
    uint oldBalance =  balances[this].add(balances[msg.sender]);
    balances[this] = balances[this].add(_amount);
    balances[msg.sender] = balances[msg.sender].sub(_amount);
    revenue = _amount.mul(sellPriceInWei).div(1 ether);
    require(revenue > 0);
    if(!msg.sender.send(revenue)){
      revert();
    }
    else {
      uint newBalance =  balances[this].add(balances[msg.sender]);
      assert(newBalance == oldBalance);
      Transfer(msg.sender, this, _amount);
      return _amount;
    }
  }

  function freezeAccount(address target, bool freeze) public onlyOwner {
    frozenAccounts[target] = freeze;
    FrozenFunds(target, freeze);
  }

  function setMinBalance(uint minimumBalanceInWei) public onlyOwner {
    minBalanceForAccounts = minimumBalanceInWei;
  }

  /* @dev send ETH to the company wallet, the token address should keep a reasonable amount of ETH to be able to payout on token sells */
  function forwardFundToCompanyWallet(uint _amount) public onlyOwner {
    companyWallet.transfer(_amount);
  }

  function distribute(address _to, uint256 _value) public onlyOwner returns (bool) {
    require(_to != address(0));
    balances[this] = balances[this].sub(_value);
    balances[_to] = balances[_to].add(_value);
    Transfer(this, _to, _value);
    return true;
  }

  // fallback function can be used to buy tokens
  function () external payable {
    buy();
  }

}